

Shifting and Scaling the Graph of a Function

Objective

This lab continues the introduction to Maple by learning how to create more complicated plots: plots with graphs of more than one function, legends, different colors and styles,

The examples and assignment will have you working with horizontal and vertical shifts and scalings.

Background

There are two fundamentally different ways to work with functions in Maple. The first method, used in Lab 1, is to enter the function as an *expression*. The following commands illustrate how a function entered as an expression is evaluated at a point, $x = 2$, and used to form the difference quotient.

Example 1: Functions as Expressions

```
> F := (x+2)*(x^2-1);           # define function as an expression
> plot( F, x=-3..3 );          # plot of function on domain [-3,3]
> eval( F, x=2 );              # compute F(2)
> dq_F := (eval( F, x=x+h ) - F)/h; # compute  $\frac{F(x+h)-F(x)}{h}$ 
> simplify( dq_F );           # perform algebraic simplification
```

The second method uses the Maple arrow operator (\rightarrow) to define a rule that maps the input, x , to a value of the function, $(x + 2)(x^2 - 1)$.

Example 2: Functions as Mappings

```
> f := x -> (x+2)*(x^2-1);     # define function as a mapping
> plot( f(x), x=-3..3 );       # plot of function on domain [-3,3]
> f(2);                         # compute f(2)
> dq_f := (f(x+h)-f(x))/h;     # compute  $\frac{f(x+h)-f(x)}{h}$ 
> simplify( dq_f );           # perform algebraic simplification
```

This example illustrates that it is easier to evaluate a function at a specific point with the second method. Expressions are easier to define but evaluation at a specific point requires the `eval` command.

Discussion

Example 3: Getting Started

```
> restart;                       # clear Maple's memory
> with( plots );                 # load additional plotting commands
> f := x -> (x+2)*(x^2-1);       # define  $f(x) = (x + 2)(x^2 - 1)$ 
```

Example 4: Horizontal and Vertical Shifts

```
> plot( [ f(x), f(x+1), f(x-2) ], # plot of function and two horizontal
>       x=-3..3, y=-20..20 );
> plot( [ f(x), f(x)+1, f(x)-2 ], # plot of function and two vertical shifts
>       x=-3..3, y=-20..20 );
```

Example 5: Animated Horizontal and Vertical Shifts

```
> animate( f(x+a), x=-8..8, a=-4..4,           # animation w/9 frames
>         frames=9, view=[DEFAULT,-20..20] );
> Pf := plot( f(x), x=-8..8, color=blue ):     # note use of colon
> Phshift := animate( f(x+a), x=-8..8,         # animation w/49 frames
>                   a=-4..4, frames=49,
>                   view=[DEFAULT,-20..20] ):
> display( [Pf, Phshift] );                    # overlay orig on anim
```

Notes

- (1) The display of multiple plots and animations could have been created with expressions; the use of `eval` would make the commands very complicated and much less clear. This is definitely a place to use mappings.
- (2) Maple uses different colors for different curves in a plot. The third curve, drawn in yellow, can be very difficult to see. The colors can be changed by using the `color=` option to the `plot` command.
- (3) A Maple plot can be copied to many other applications, e.g., Word or PowerPoint. A Maple plot pasted into another application cannot be edited. All changes to the plot must be made within Maple.
- (4) If a plot is to be printed on a black and white printer, colors are not a good way to identify curves. In this situation the `linestyle=` option can be used.
- (5) By default Maple uses 50 points to plot a function. If this produces a jagged graph for a smooth function, include the `numpoints=` argument in the `plot` command.
- (6) To see a plot of points instead of the curve, use `style=point`. To distinguish multiple sets of points, use `color=` and/or `symbol=`. A list of 25 pre-defined colors can be found on the `plot,color` help page (accessible via the command `?plot,color`).
- (7) The `title=` and `legend=` options put a title and legend, respectively, on a Maple plot. A legend can also be inserted interactively. To access this feature, position the cursor above a Maple plot and click the right mouse button. Select **Legend, Edit Legend ...**
- (8) For additional information about the optional arguments for the `plot` command, see the help page named `plot,options`.

Questions

Let $f(x) = (x + 2)(x^2 - 1)$ on the domain $[-3, 3]$.

- (1) Create a Maple plot containing the graph of $y = f(x)$ together with the graphs of the horizontal shifts to the right by 2, to the right by 1, and to the left by 1. Use colors that are easily visible, i.e., not yellow. Include a legend that clearly identifies each of the four curves in the plot. *Suggestion:* Copy your plot into a Word document.
- (2) Create a Maple plot containing the graph of $y = f(x)$ together with the graphs of the vertical shifts down by 5 and up by 2. Again, use easily visible colors and include a legend identifying the three curves in the plot.
- (3) A horizontal scaling of the graph of $y = f(x)$ is the graph of $y = f(ax)$ where a is referred to as the scale factor of the scaling. What Maple command will create a 21-frame animation of the horizontal scaling of the graph of $y = f(x)$ with scale factors ranging from -10 to 10? *Be sure you specify a plot window that makes a good display for all frames in the animation. It might be necessary to use more than 50 points to create smooth curves for all scale factors.*
- (4) One of the horizontal scalings of the graph of $y = f(x)$ does not cross the x -axis. Which one? Why?